

# NAG Library Routine Document

## G07DBF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G07DBF computes an  $M$ -estimate of location with (optional) simultaneous estimation of the scale using Huber's algorithm.

### 2 Specification

```

SUBROUTINE G07DBF (ISIGMA, N, X, IPSI, C, H1, H2, H3, DCHI, THETA,      &
                  SIGMA, MAXIT, TOL, RS, NIT, WRK, IFAIL)
INTEGER          ISIGMA, N, IPSI, MAXIT, NIT, IFAIL
REAL (KIND=nag_wp) X(N), C, H1, H2, H3, DCHI, THETA, SIGMA, TOL, RS(N), &
                  WRK(N)

```

### 3 Description

The data consists of a sample of size  $n$ , denoted by  $x_1, x_2, \dots, x_n$ , drawn from a random variable  $X$ . The  $x_i$  are assumed to be independent with an unknown distribution function of the form

$$F((x_i - \theta)/\sigma)$$

where  $\theta$  is a location argument, and  $\sigma$  is a scale argument.  $M$ -estimators of  $\theta$  and  $\sigma$  are given by the solution to the following system of equations:

$$\sum_{i=1}^n \psi\left(\frac{x_i - \hat{\theta}}{\hat{\sigma}}\right) = 0 \quad (1)$$

$$\sum_{i=1}^n \chi\left(\frac{x_i - \hat{\theta}}{\hat{\sigma}}\right) = (n - 1)\beta \quad (2)$$

where  $\psi$  and  $\chi$  are given functions, and  $\beta$  is a constant, such that  $\hat{\sigma}$  is an unbiased estimator when  $x_i$ , for  $i = 1, 2, \dots, n$  has a Normal distribution. Optionally, the second equation can be omitted and the first equation is solved for  $\hat{\theta}$  using an assigned value of  $\sigma = \sigma_c$ .

The values of  $\psi\left(\frac{x_i - \hat{\theta}}{\hat{\sigma}}\right)\hat{\sigma}$  are known as the Winsorized residuals.

The following functions are available for  $\psi$  and  $\chi$  in G07DBF.

#### (a) Null Weights

$$\psi(t) = t \qquad \chi(t) = \frac{t^2}{2}$$

Use of these null functions leads to the mean and standard deviation of the data.

#### (b) Huber's Function

$$\psi(t) = \max(-c, \min(c, t)) \qquad \chi(t) = \frac{\|t\|^2}{2} \|t\| \leq d$$

$$\chi(t) = \frac{d^2}{2} \|t\| > d$$

**(c) Hampel's Piecewise Linear Function**

$$\psi_{h_1, h_2, h_3}(t) = -\psi_{h_1, h_2, h_3}(-t)$$

$$\psi_{h_1, h_2, h_3}(t) = t \quad 0 \leq t \leq h_1 \quad \chi(t) = \frac{|t|^2}{2}|t| \leq d$$

$$\psi_{h_1, h_2, h_3}(t) = h_1 \quad h_1 \leq t \leq h_2$$

$$\psi_{h_1, h_2, h_3}(t) = h_1(h_3 - t)/(h_3 - h_2) \quad h_2 \leq t \leq h_3 \quad \chi(t) = \frac{d^2}{2}|t| > d$$

$$\psi_{h_1, h_2, h_3}(t) = 0 \quad t > h_3$$

**(d) Andrew's Sine Wave Function**

$$\psi(t) = \sin t \quad -\pi \leq t \leq \pi \quad \chi(t) = \frac{|t|^2}{2}|t| \leq d$$

$$\psi(t) = 0 \quad \text{otherwise} \quad \chi(t) = \frac{d^2}{2}|t| > d$$

**(e) Tukey's Bi-weight**

$$\psi(t) = t(1 - t^2)^2 \quad |t| \leq 1 \quad \chi(t) = \frac{|t|^2}{2}|t| \leq d$$

$$\psi(t) = t(1 - t^2)^2 = 0 \quad \text{otherwise} \quad \chi(t) = \frac{d^2}{2}|t| > d$$

where  $c$ ,  $h_1$ ,  $h_2$ ,  $h_3$  and  $d$  are constants.

Equations (1) and (2) are solved by a simple iterative procedure suggested by Huber:

$$\hat{\sigma}_k = \sqrt{\frac{1}{\beta(n-1)} \left( \sum_{i=1}^n \chi \left( \frac{x_i - \hat{\theta}_{k-1}}{\hat{\sigma}_{k-1}} \right) \right) \hat{\sigma}_{k-1}^2}$$

and

$$\hat{\theta}_k = \hat{\theta}_{k-1} + \frac{1}{n} \sum_{i=1}^n \psi \left( \frac{x_i - \hat{\theta}_{k-1}}{\hat{\sigma}_k} \right) \hat{\sigma}_k$$

or

$$\hat{\sigma}_k = \sigma_c, \quad \text{if } \sigma \text{ is fixed.}$$

The initial values for  $\hat{\theta}$  and  $\hat{\sigma}$  may either be user-supplied or calculated within G07DBF as the sample median and an estimate of  $\sigma$  based on the median absolute deviation respectively.

G07DBF is based upon subroutine LYHALG within the ROBETH library, see Marazzi (1987).

**4 References**

Hampel F R, Ronchetti E M, Rousseeuw P J and Stahel W A (1986) *Robust Statistics. The Approach Based on Influence Functions* Wiley

Huber P J (1981) *Robust Statistics* Wiley

Marazzi A (1987) Subroutines for robust estimation of location and scale in ROBETH *Cah. Rech. Doc. IUMSP, No. 3 ROB 1* Institut Universitaire de Médecine Sociale et Préventive, Lausanne

## 5 Arguments

- 1: ISIGMA – INTEGER *Input*  
*On entry:* the value assigned to ISIGMA determines whether  $\hat{\sigma}$  is to be simultaneously estimated.  
 ISIGMA = 0  
     The estimation of  $\hat{\sigma}$  is bypassed and SIGMA is set equal to  $\sigma_c$ .  
 ISIGMA = 1  
      $\hat{\sigma}$  is estimated simultaneously.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of observations.  
*Constraint:*  $N > 1$ .
- 3: X(N) – REAL (KIND=nag\_wp) array *Input*  
*On entry:* the vector of observations,  $x_1, x_2, \dots, x_n$ .
- 4: IPSI – INTEGER *Input*  
*On entry:* which  $\psi$  function is to be used.  
 IPSI = 0  
      $\psi(t) = t$ .  
 IPSI = 1  
     Huber's function.  
 IPSI = 2  
     Hampel's piecewise linear function.  
 IPSI = 3  
     Andrew's sine wave,  
 IPSI = 4  
     Tukey's bi-weight.
- 5: C – REAL (KIND=nag\_wp) *Input*  
*On entry:* if IPSI = 1, C must specify the argument,  $c$ , of Huber's  $\psi$  function. C is not referenced if IPSI  $\neq$  1.  
*Constraint:* if IPSI = 1,  $C > 0.0$ .
- 6: H1 – REAL (KIND=nag\_wp) *Input*  
 7: H2 – REAL (KIND=nag\_wp) *Input*  
 8: H3 – REAL (KIND=nag\_wp) *Input*  
*On entry:* if IPSI = 2, H1, H2 and H3 must specify the arguments,  $h_1, h_2$ , and  $h_3$ , of Hampel's piecewise linear  $\psi$  function. H1, H2 and H3 are not referenced if IPSI  $\neq$  2.  
*Constraint:*  $0 \leq H1 \leq H2 \leq H3$  and  $H3 > 0.0$  if IPSI = 2.
- 9: DCHI – REAL (KIND=nag\_wp) *Input*  
*On entry:*  $d$ , the argument of the  $\chi$  function. DCHI is not referenced if IPSI = 0.  
*Constraint:* if IPSI  $\neq$  0,  $DCHI > 0.0$ .

- 10: THETA – REAL (KIND=nag\_wp) Input/Output  
*On entry:* if SIGMA > 0 then THETA must be set to the required starting value of the estimation of the location argument  $\hat{\theta}$ . A reasonable initial value for  $\hat{\theta}$  will often be the sample mean or median.  
*On exit:* the  $M$ -estimate of the location argument,  $\hat{\theta}$ .
- 11: SIGMA – REAL (KIND=nag\_wp) Input/Output  
*On entry:* the role of SIGMA depends on the value assigned to ISIGMA, as follows:  
 if ISIGMA = 1, SIGMA must be assigned a value which determines the values of the starting points for the calculations of  $\hat{\theta}$  and  $\hat{\sigma}$ . If SIGMA  $\leq$  0.0 then G07DBF will determine the starting points of  $\hat{\theta}$  and  $\hat{\sigma}$ . Otherwise the value assigned to SIGMA will be taken as the starting point for  $\hat{\sigma}$ , and THETA must be assigned a value before entry, see above;  
 if ISIGMA = 0, SIGMA must be assigned a value which determines the value of  $\sigma_c$ , which is held fixed during the iterations, and the starting value for the calculation of  $\hat{\theta}$ . If SIGMA  $\leq$  0, then G07DBF will determine the value of  $\sigma_c$  as the median absolute deviation adjusted to reduce bias (see G07DAF) and the starting point for  $\hat{\theta}$ . Otherwise, the value assigned to SIGMA will be taken as the value of  $\sigma_c$  and THETA must be assigned a relevant value before entry, see above.  
*On exit:* contains the  $M$ -estimate of the scale argument,  $\hat{\sigma}$ , if ISIGMA was assigned the value 1 on entry, otherwise SIGMA will contain the initial fixed value  $\sigma_c$ .
- 12: MAXIT – INTEGER Input  
*On entry:* the maximum number of iterations that should be used during the estimation.  
*Suggested value:* MAXIT = 50.  
*Constraint:* MAXIT > 0.
- 13: TOL – REAL (KIND=nag\_wp) Input  
*On entry:* the relative precision for the final estimates. Convergence is assumed when the increments for THETA, and SIGMA are less than TOL  $\times$  max(1.0,  $\sigma_{k-1}$ ).  
*Constraint:* TOL > 0.0.
- 14: RS(N) – REAL (KIND=nag\_wp) array Output  
*On exit:* the Winsorized residuals.
- 15: NIT – INTEGER Output  
*On exit:* the number of iterations that were used during the estimation.
- 16: WRK(N) – REAL (KIND=nag\_wp) array Output  
*On exit:* if SIGMA  $\leq$  0.0 on entry, WRK will contain the  $n$  observations in ascending order.
- 17: IFAIL – INTEGER Input/Output  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N \leq 1$ ,  
 or  $\text{MAXIT} \leq 0$ ,  
 or  $\text{TOL} \leq 0.0$ ,  
 or  $\text{ISIGMA} \neq 0$  or 1,  
 or  $\text{IPSI} < 0$ ,  
 or  $\text{IPSI} > 4$ .

IFAIL = 2

On entry,  $C \leq 0.0$  and  $\text{IPSI} = 1$ ,  
 or  $H1 < 0.0$  and  $\text{IPSI} = 2$ ,  
 or  $H1 = H2 = H3 = 0.0$  and  $\text{IPSI} = 2$ ,  
 or  $H1 > H2$  and  $\text{IPSI} = 2$ ,  
 or  $H1 > H3$  and  $\text{IPSI} = 2$ ,  
 or  $H2 > H3$  and  $\text{IPSI} = 2$ ,  
 or  $\text{DCHI} \leq 0.0$  and  $\text{IPSI} \neq 0$ .

IFAIL = 3

On entry, all elements of the input array X are equal.

IFAIL = 4

SIGMA, the current estimate of  $\sigma$ , is zero or negative. This error exit is very unlikely, although it may be caused by too large an initial value of SIGMA.

IFAIL = 5

The number of iterations required exceeds MAXIT.

IFAIL = 6

On completion of the iterations, the Winsorized residuals were all zero. This may occur when using the ISIGMA = 0 option with a redescending  $\psi$  function, i.e., Hampel's piecewise linear function, Andrew's sine wave, and Tukey's biweight.

If the given value of  $\sigma$  is too small, then the standardized residuals  $\frac{x_i - \hat{\theta}_k}{\sigma_c}$ , will be large and all the residuals may fall into the region for which  $\psi(t) = 0$ . This may incorrectly terminate the iterations thus making THETA and SIGMA invalid.

Re-enter the routine with a larger value of  $\sigma_c$  or with ISIGMA = 1.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

On successful exit the accuracy of the results is related to the value of TOL, see Section 5.

## 8 Parallelism and Performance

G07DBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

G07DBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

When you supply the initial values, care has to be taken over the choice of the initial value of  $\sigma$ . If too small a value of  $\sigma$  is chosen then initial values of the standardized residuals  $\frac{x_i - \hat{\theta}_k}{\sigma}$  will be large. If the redescending  $\psi$  functions are used, i.e., Hampel's piecewise linear function, Andrew's sine wave, or Tukey's bi-weight, then these large values of the standardized residuals are Winsorized as zero. If a sufficient number of the residuals fall into this category then a false solution may be returned, see page 152 of Hampel *et al.* (1986).

## 10 Example

The following program reads in a set of data consisting of eleven observations of a variable  $X$ .

For this example, Hampel's Piecewise Linear Function is used (IPSI = 2), values for  $h_1$ ,  $h_2$  and  $h_3$  along with  $d$  for the  $\chi$  function, being read from the data file.

Using the following starting values various estimates of  $\theta$  and  $\sigma$  are calculated and printed along with the number of iterations used:

- (a) G07DBF determines the starting values,  $\sigma$  is estimated simultaneously.
- (b) You must supply the starting values,  $\sigma$  is estimated simultaneously.
- (c) G07DBF determines the starting values,  $\sigma$  is fixed.
- (d) You must supply the starting values,  $\sigma$  is fixed.

**10.1 Program Text**

```

Program g07dbfe

!      G07DBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: g07dbf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: c, dchi, h1, h2, h3, sigma, sigsav, &
                           thesav, theta, tol
Integer                    :: ifail, ipsi, isigma, maxit, n, nit
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: rs(:), wrk(:), x(:)
!      .. Executable Statements ..
Write (nout,*) 'G07DBF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the problem size
Read (nin,*) n

Allocate (x(n),rs(n),wrk(n))

!      Read in data
Read (nin,*) x(1:n)

!      Read in details of algorithm to use
Read (nin,*) ipsi, dchi, maxit
If (ipsi==2) Then
  Read (nin,*) h1, h2, h3
End If

!      Display titles
Write (nout,*) '          Input parameters      Output parameters'
Write (nout,*) 'ISIGMA  SIGMA  THETA  TOL  SIGMA  THETA'

d_lp: Do
  Read (nin,*,Iostat=ifail) isigma, sigma, theta, tol
  If (ifail/=0) Then
    Exit d_lp
  End If

!      Save the input parameters for later display
  sigsav = sigma
  thesav = theta

!      Compute M-estimates
  ifail = 0
  Call g07dbf(isigma,n,x,ipsi,c,h1,h2,h3,dchi,theta,sigma,maxit,tol,rs, &
             nit,wrk,ifail)

!      Display results
  Write (nout,99999) isigma, sigsav, thesav, tol, sigma, theta
End Do d_lp

99999 Format (1X,I3,3X,2F8.4,F7.4,F9.4,F8.4,I4)
End Program g07dbfe

```

## 10.2 Program Data

```
G07DBF Example Program Data
11                                :: N
13.0 11.0 16.0 5.0 3.0 18.0
9.0 8.0 6.0 27.0 7.0           :: End of X
2 1.5 50                        :: IPSI,DCHI,MAXIT
1.5 3.0 4.5                      :: H1,H2,H3
1 -1.0 0.0 0.0001              :: ISIGMA,SIGMA,THETA,TOL
1 7.0 2.0 0.0001               :: ISIGMA,SIGMA,THETA,TOL
0 -1.0 0.0 0.0001             :: ISIGMA,SIGMA,THETA,TOL
0 7.0 2.0 0.0001              :: ISIGMA,SIGMA,THETA,TOL
```

## 10.3 Program Results

G07DBF Example Program Results

ISIGMA	Input parameters			Output parameters	
	SIGMA	THETA	TOL	SIGMA	THETA
1	-1.0000	0.0000	0.0001	6.3247	10.5487
1	7.0000	2.0000	0.0001	6.3249	10.5487
0	-1.0000	0.0000	0.0001	5.9304	10.4896
0	7.0000	2.0000	0.0001	7.0000	10.6500

---